
本記事では、Java 8とJava 9でJVMがメソッドを実行する方法について説明します。このトピックは、JVMの内

部に関する基本的な内容です。JVMのJust-In-Time（JIT）コンパイラを理解したい方や、アプリケーション

の実行をチューニングしたい方にとって、欠かせない背景知識です。

バイトコードの内部
まずは、簡単なJavaコードを見てみます。

long time = System.currentTimeMillis();
HashMap<String, String> hm = new HashMap<>();
hm.put("now", "bar");
Map<String, String> m = hm;
m.put("foo", "baz");

Javaコンパイラがこのコードから生成したバイトコードを確認するためには、-cスイッチを付けてjavapツールを使

用します。実行すると、逆コンパイルされたコードが表示されます。

0: invokestatic #2 // Method  
                     java/lang/System.currentTimeMillis:()J
3: lstore_1
4: new #3 // class java/util/HashMap
7: dup写真：JOHN BLYTHE

Javaにおけるメソッド 
呼出しの仕組み
特殊なバイトコードでメソッドの呼出しを大幅に効率化。その仕組みを理解
することで、JVMがコードを実行する方法を解き明かす

8: invokespecial #4 // Method java/util/HashMap."<init>":()V
11: astore_3
12: aload_3
13: ldc #5 // String now
15: ldc #6 // String bar
17: invokevirtual #7 // Method java/util/HashMap.put: 
                     (Ljava/lang/Object; Ljava/lang/Object;) 
                     Ljava/lang/Object;
20: pop
21: aload_3
22: astore 4
24: aload 4
26: ldc #8 // String foo
28: ldc #9 // String baz
30: invokeinterface #10, 3 // InterfaceMethod java/util/Map.put:
                     (Ljava/lang/Object;Ljava/lang/Object;) 
                     Ljava/lang/Object;
35: pop

[編集注：インデントされた行は、前の行の続きです。] 実際のJavaのメソッド呼出しは、何種類かある、invoke*と
いう形式のバイトコードの1つに変換されます。JVMレベルのコードを見るのは初めてだというJavaプログラマーの

方は、このことを知って驚かれるかもしれません。 

それでは、逆コンパイルされたコードの最初の部分を詳しく見てみます。

0: invokestatic #2 // Method java/lang/System.currentTimeMillis:()J
3: lstore_1

System.currentTimeMillis()メソッドの静的呼出しは、invokestatic命令コードに変換され、バイトコードの0の位

置に格納されています。このメソッドにはパラメータがないため、呼出し前に評価スタックにロードするものはあり

ません。



ORACLE.COM/JAVAMAGAZINE  ////////////////////  SEPTEMBER/OCTOBER 2017

32

//java 9 /

次に、バイト・ストリームに2つのバイト00 02が登場します。この2つは組み合わされて16ビット数値（この場合

は#2）となり、クラス・ファイル内の表（この表を定数プールと呼びます）に対するオフセットとして使用されます。定

数プールのインデックスはすべて16ビットであるため、命令コードがプール内のエントリを参照する場合は常に、エ

ントリのオフセットをエンコードした連続する2バイトが必要です。

便利なことに、逆コンパイラはオフセット#2に対応するメソッドが何かを示すコメントを追加します。今回の

例では、予想どおりSystem.currentTimeMillis()メソッドとなっています。javapでは、逆コンパイルされた出力の

中に、呼び出されたメソッドの名前、メソッドが受け取るパラメータの型（括弧内）、メソッドの戻り型が表示され

ます。

呼出し先から戻る際には、呼出しの結果がスタックに配置されます。オフセット3の位置には、引数のない単

独の命令コードlstore_1があります。この処理は、戻り値をlong型のローカル変数に保存するものです。

もちろん、人間が読めば、この値は二度と使われないことを理解できます。しかし、Javaコンパイラにおける

設計上の目標の1つは、論理的に意味をなすかどうかにかかわらず、Javaソース・コードの内容をできる限り忠実に

表現する点にあります。そのため、System.currentTimeMillis()の戻り値は、この後のプログラムで使われることが

なくても保存されます。

それでは、逆コンパイルされたコードの次の部分を見てみます。

4: new #3 // class java/util/HashMap
7: dup
8: invokespecial #4 // Method java/util/HashMap."<init>":()V
11: astore_3
12: aload_3
13: ldc #5 // String now
15: ldc #6 // String bar
17: invokevirtual #7 // Method java/util/HashMap.put:
                        (Ljava/lang/Object;Ljava/lang/Object;)
                        Ljava/lang/Object;
20: pop

バイトコード4から10では、新しいHashMapのインスタンスを作成しています。その後の命令11では、そのコピーを

ローカル変数に保存しています。次に、命令12から16で、HashMapオブジェクトと、put()の呼出しに使う引数が配

置されたスタックを準備しています。put()メソッドの実際の呼出しは、命令17から19で行っています。

今回使われているinvoke命令コードは、invokevirtualです。これは、メソッド呼出しの対象となるインスタン

ス（レシーバ・オブジェクトと呼ばれることもあります）が存在しない静的呼出しとは異なります（インスタンスを

呼び出すバイトコードでは、レシーバと、呼出しに使うすべての引数を評価スタックに配置してから、invoke命令を

発行する必要があります）。今回の例では、put()の戻り値は使われていないため、命令20で戻り値を破棄していま

す。

21から25までのバイト列は、最初は少し奇妙に見えるかもしれません。 

21: aload_3
22: astore 4
24: aload 4
26: ldc #8 // String foo
28: ldc #9 // String baz
30: invokeinterface #10, 3 // InterfaceMethod java/util/Map.put:
                             (Ljava/lang/Object;Ljava/lang/Object;)
                             Ljava/lang/Object;
35: pop

バイトコード4で作成され、命令11でローカル変数3に保存されたHashMapインスタンスが、ここでスタック上に再

ロードされ、その参照のコピーがローカル変数4に保存されています。この処理によって、インスタンスがスタックか

ら削除されるため、このインスタンスを使う前に（変数4から）再ロードする必要があります。

このような変数のシャッフルが起こるのは、元のJavaコードで、追加のローカル変数（HashMap型ではなく

Map型）が作成されているためです。ただし、この変数は常に元の変数と同じオブジェクトを参照しています。これ

は、バイトコードが元のソース・コードにできる限り近くなっているもう1つの例と言えます。Javaがこの「単純なバイ

トコード」と言われるアプローチを採用している主な理由の1つは、JVMのJITコンパイラへの入力フォーマットを簡

単なものにすることです。

スタックと変数がシャッフルされた後、命令26から29でMap内に配置される値がロードされます。これでレシ

ーバと引数が格納でき、スタックの準備ができました。続く命令30では、put()が呼び出されます。今回の命令コー

ドはinvokeinterfaceですが、実際に呼び出されるメソッドは同じです。ここでも、命令35のpopでput()の戻り値

は破棄されています。

ここまでの部分で、Javaコンパイラは呼出しの状況に応じてinvokestatic、invokevirtua
l、invokeinterfaceを生成し分けていることがおわかりいただけたかと思います。 

メソッド呼出しに使われるJVMバイトコード
それでは、メソッド呼出しに使われる可能性がある5つのJVMバイトコードすべてについて見ていきます（表1参照）。

いずれのケースでも、バイトコードb0とb1が組み合わされて定数プールのオフセットc1となります。

実際にJavaコードを書いて生成されたJavaクラスをjavapで逆アセンブルし、それぞれの形式のバイトコード

がどのような状況で生成されるのかを確認してみると、よい演習になるでしょう。

もっとも一般的なメソッド呼出しは、仮想ディスパッチを示すinvokevirtualです。仮想ディスパッチという用語

は、呼び出されるメソッドが実行時に完全に決定されることを意味しています。これを理解するためには、実行中

のアプリケーション内に存在する各クラスに対して、その型に対応するメタデータを格納するメモリ領域がJVM内

に確保されていることを知る必要があります。この領域は、（少なくとも、Java HotSpot VMでは）klassと呼ばれて

おり、JVMでの型情報の表現と考えることができます。

Java 7以前では、klassメタデータは、permgenと呼ばれるJavaヒープに格納されていました。Javaヒープ内の

表1：メソッド呼出しに使われるJVMバイトコード

命令コード名 引数 説明
invokevirtual b0 b1 CP#C1に見つかったメソッドを仮想ディスパッチで呼び出す

invokespecial b0 b1 CP#C1に見つかったメソッドを「特殊」な（正確な）ディス
パッチで呼び出す

invokeinterface b0 b1 x0 00 CP#C1に見つかったインタフェース・メソッドをインタフェー
ス・オフセット検索で呼び出す

invokestatic b0 b1 CP#C1に見つかった静的メソッドを呼び出す

invokedynamic b0 b1 00 00 呼び出すメソッドを動的に検索して呼び出す

オブジェクトには、必ずオブジェクト・ヘッダー（oopと呼ばれています）が存在するため、klassはklassOopと呼ば

れていました。Java 8とJava 9では、klassメタデータがJavaヒープからネイティブ・ヒープに移動したため、オブジェ

クト・ヘッダーは不要になっています。Javaプログラマーは、型に対応するClass<?>オブジェクトを使ってklassの情

報の一部を参照できますが、これは別の概念です。

klass内の領域で特に重要なのがvtableです。実質的に、この領域は、その型で定義されているメソッドの実装

を指す関数ポインタの表だと言えます。invokevirtualでインスタンス・メソッドが呼び出されると、JVMはvtableを

参照して、実行する必要があるコードを正確に特定します。klassにメソッドの定義がない場合、JVMはスーパークラ

スに対応するklassへのポインタをたどり、再度呼出しを試みます。

これがJVMでのメソッドのオーバーライド処理の基本となりますが、この処理を効率的に行うことができるよ

うに、vtableは決まった方法でレイアウトされています。各klassのvtableは、親の型で定義されているメソッドが最

初に来るようになっており、親の型とまったく同じ順番でレイアウトされます。親クラスで宣言されておらず、その型

で新しく追加されたメソッドはvtableの最後に配置されます。

つまり、サブクラスでメソッドをオーバーライドした場合、オーバーライドした実装とオーバーライドされた実

装でvtableのオフセットが同じ値になります。この仕組みによって、vtableのオフセットが親のオフセットと同じにな

るため、オーバーライドされたメソッドのルックアップがとても簡単になります。

図1は、Pet、Cat、Bearの各クラスとFurryインタフェースが定義されている例です。

図1：単純な継承階層

Bear

Object

groom()

groom()

Furry

Pet feed()

Cat groom()
purr()

Java 7では、これらのクラスのvtableは図2のようなレイアウトになります。なお、この図はJava 7のpermgen

内のレイアウトを示しているため、参照先はklassOopとなっています。そのため、2ワードのオブジェクト・ヘッダー（

図では、mおよびkkで示されています）が存在します。前述のとおり、Java 8とJava 9ではこのエントリは存在しま

せんが、その点を除けば、同じ図が適用されます。

Cat::feedを呼び出す場合を考えてみます。JVMは、Catクラスでオーバーライドされたメソッドを見つけられ

ないため、Petのklassへのポインタをたどります。このklassにはfeed()の実装が存在するため、そのコードが呼び出

されることになります。このvtable構造がうまく動作するのは、Javaがクラスの単一継承しか実装していないためで

す。つまり、どの型にも、直接的なスーパークラスは1つしかありません（スーパークラスのないオブジェクトは例外

です）。

invokeinterfaceが使われる場合は、もう少し複雑になります。たとえば、すべてのFurryの実装でgroom()メ
ソッドが必ずしもvtableの同じ場所にあるとは限りません。Cat::groomとBear::groomでオフセットが異なるの

は、それぞれのクラスの継承階層が異なるためです。そのため、コンパイル時にインタフェース型しかわからないオ

ブジェクトのメソッドを呼び出す場合、追加でルックアップを行う必要があります。

インタフェースの呼出しの際に、追加のルックアップ作業が若干必要にはなりますが、インタフェースの使用を

図2： 図1に示したクラスのvtableの構造

klassOop
(object)

toString()

toString() feed()

groom()

purr() groom()

klassOop
(Pet)

klassOop
(Bear)

避けて微少な最適化を試みるべきではないという点に注意してください。JVMにはJITコンパイラが搭載されている

ため、この2つのケースに実質的なパフォーマンスの差はないことを覚えておいてください。

メソッド呼出しの例
別の例を見てみます。次のコードについて考えてみます。

Cat tom = new Cat();
Bear pooh = new Bear();
Furry f;

tom.groom();
pooh.groom();

f = tom;
f.groom();

f = pooh;
f.groom();

このコードから、次のバイトコードが生成されます。

Code:
0: new           #2            // class scratch/Cat
3: dup
4: invokespecial #3            // Method scratch/Cat."<init>":()V
7: astore_1
8: new           #4            // class scratch/Bear
11: dup
12: invokespecial #5           // Method scratch/Bear."<init>":()V
15: astore_2
16: aload_1

17: invokevirtual #6           // Method scratch/Cat.groom:()V
20: aload_2
21: invokevirtual #7           // Method scratch/Bear.groom:()V
24: aload_1
25: astore_3
26: aload_3
27: invokeinterface #8,  1     // InterfaceMethod scratch/Furry.groom:()V
32: aload_2
33: astore_3
34: aload_3
35: invokeinterface #8,  1     // InterfaceMethod scratch/Furry.groom:()V

27と35で行われている2つの呼出しは同じように見えますが、実際には異なるメソッドを呼び出しています。27の
呼出しではCat::groomが、35の呼出しではBear::groomが呼ばれます。

これまでに説明したinvokevirtualとinvokeinterfaceを知っておけば、invokespecialの動作が理解しやすく

なるでしょう。invokespecialでメソッドが呼び出されると、JVMは仮想ルックアップを実行せずに、リクエストされ

たメソッドの正確な場所のみをvtable上で検索します。つまり、invokespecialは、privateメソッドの呼出し、スーパ

ークラスのメソッドの呼出し、コンストラクタ本体の呼出し（バイトコードでは、<init>というメソッドに変換されま

す）という3つの場合に使用されます。3つの場合とも、オーバーライドされる可能性が明示的に排除されており、

仮想ルックアップは必要ありません。

finalメソッド
もう1つ、触れておくべき特殊なケースが、finalメソッドです。一見したところでは、finalメソッドの呼出しも

invokespecial命令に変換されると思われるかもしれません。しかし、Java言語仕様のセクション13.4.17には、final

メソッドについて「finalと宣言されたメソッドをfinalと宣言されていないメソッドに変更しても、既存のバイナリと

の互換性は失われない」と述べられています。

それでは、コンパイラがfinalメソッドの呼出しをinvokespecialにコンパイルした場合を考えてみます。メソッ

ドがfinal以外に変更されると、サブクラスでオーバーライドされる可能性があります。ここで、コンパイルされたコ

ードにサブクラスのインスタンスが渡されたとします。すると、invokespecialが実行され、誤ったメソッドの実装が

javap
JavaのJVM命令セット

learn more

呼び出されることになります。これは、Javaのオブジェクト指向のルールに違反します（厳密に言えば、リスコフの

置換原則に違反することになります）。

そのため、finalメソッドの呼出しはinvokevirtual命令にコンパイルされる必要があります。実際のJavaの

HotSpot VMは、finalメソッドを検出して非常に効率よく実行できるように最適化されています。

まとめ
ここまでで、JVMがサポートする5つの呼出し命令のうち4つを説明してきました。残るinvokedynamicは、

これだけで1本の記事が必要になるほど奥深く興味深いテーマです。2回シリーズの2回目となる次の記事で

は、invokedynamicとそれに関連する題材を扱いたいと思います。</article>

